78 CS 536 Fall 2002^©

Java CUP

Java CUP is a parser-generation tool, similar to Yacc. CUP builds a Java parser for LALR(1) grammars from production rules and associated Java code fragments. When a particular production is recognized, its associated code fragment is executed (typically to build an AST). CUP generates a Java source file

parser.java. It contains a class
parser, with a method
 Symbol parse()

TheSymbol returned by the parser is associated with the grammar’s start symbol and contains the AST for the whole source program. 79 CS 536 Fall 2002^©

Thefile sym.java isalsobuiltforuse with a JLex-built scanner (so that bothscannerandparserusethe same token codes). Ifanunrecoveredsyntaxerroroccurs,

Exception() isthrownbythe parser.

CUPandYaccacceptexactlythesame class of grammars—all LL(1) grammars, plus many useful non- LL(1) grammars. CUP is called as

java java_cup.Main < file.cup

80 CS 536 Fall 2002^©

Java CUP Specifications

Java CUP specifications are of the form:

• Package and import specifications

• User code additions

• Terminal and non-terminal

declarations

• A context-free grammar, augmented

with Java code fragments

Package and Import Specifications You define a package name as:

package name ;

You add imports to be used as:

import java_cup.runtime.*;

81 CS 536 Fall 2002^©

User Code Additions You may define Java code to be included within the generated parser:

action code {: /*java code */ :}

This code is placed within the generated action class (which holds user-specified production actions).

parser code {: /*java code */ :}

This code is placed within the generated parser class.

init with{: /*java code */ :}

This code is used to initialize the generated parser.

scan with{: /*java code */ :}

Thiscodeisusedto tellthegenerated parser how to get tokens from the scanner.

82 CS 536 Fall 2002^©

Terminal and Non-terminal

Declarations You define terminal symbols you will use as:

terminal classname name_1, name_2, ...

classname is a class used by the

scanner for tokens (CSXToken,

CSXIdentifierToken, etc.)

Youdefinenon-terminalsymbolsyou will use as:

non terminal classname name_1, name_2, ...

classname is the class for the AST

node associated with the nonterminal (stmtNode, exprNode, etc.)

83 CS 536 Fall 2002^©

Production Rules Production rules are of the form

name ::= name_1 name_2 ... action ;

or

name ::= name_1 name_2 ... action_1
   | name_3 name_4 ... action_2
   | ...
   ;

Names are the names of terminals or non-terminals, as declared earlier. Actions are Java code fragments, of the form

{: /*java code */ :}

The Java object associated with a symbol(a tokenorASTnode)maybe named by adding a:id suffix to a terminal or non-terminal in a rule.

RESULT names the left-hand side

non-terminal.

84 CS 536 Fall 2002^©

The Java classes of the symbols are defined in the terminal and nonterminal declaration sections. For example,

prog ::= LBRACE:l stmts:s RBRACE
   {: RESULT=
       new csxLiteNode(s,
         l.linenum,l.colnum); :}

This corresponds to the production prog → { stmts } The left brace is given the namel; the stmts non-terminal is calleds. In the action code, a new

CSXLiteNode iscreatedandassigned

to prog. It is constructed from the AST node associated withs. Its line and column numbers are those given to the left brace,l (by the scanner).

85 CS 536 Fall 2002^©

To tellCUPwhatnon-terminalto use as the start symbol (prog in our example), we use the directive:

start with prog;

86 CS 536 Fall 2002^©

Example

Let’slookat the CUPspecificationfor CSX-lite. Recall its CFG is program→ { stmts } stmts→ stmt stmts | λ stmt→ id = expr ;

| if ( expr ) stmt expr → expr + id

| expr - id | id

87 CS 536 Fall 2002^©

The corresponding CUP specification is:

/***
This Is A Java CUP Specification For
CSX-lite, a Small Subset
of The CSX Language,  Used In Cs536
 ***/

/* Preliminaries to set up and use
the scanner.  */

import java_cup.runtime.*;
parser code {:
 public void syntax_error
  (Symbol cur_token){
   report_error(
    “CSX syntax error at line “+
    String.valueOf(((CSXToken)
       cur_token.value).linenum),
    null);}
:};

init with {:              :};
scan with {:
   return Scanner.next_token();
:};

88 CS 536 Fall 2002^©

/* Terminals (tokens returned by the
scanner). */
terminal CSXIdentifierToken
IDENTIFIER;
terminal CSXToken SEMI, LPAREN,
RPAREN, ASG, LBRACE, RBRACE;
terminal CSXToken PLUS, MINUS,
rw_IF;

/* Non terminals */
non terminal csxLiteNode prog;
non terminal stmtsNode stmts;
non terminal stmtNode stmt;
non terminal exprNode exp;
non terminal nameNode ident;

start with prog;

prog::= LBRACE:l stmts:s RBRACE
 {: RESULT=
     new csxLiteNode(s,
          l.linenum,l.colnum); :}
;

89 CS 536 Fall 2002^©

stmts::= stmt:s1  stmts:s2
 {: RESULT=
     new stmtsNode(s1,s2,
       s1.linenum,s1.colnum);
 :}
|
 {: RESULT= stmtsNode.NULL; :}
;
stmt::= ident:id ASG exp:e SEMI
 {: RESULT=
       new asgNode(id,e,
           id.linenum,id.colnum);
 :}

| rw_IF:i LPAREN exp:e RPAREN  stmt:s
 {: RESULT=new ifThenNode(e,s,
            stmtNode.NULL,
            i.linenum,i.colnum); :}
;
exp::=
 exp:leftval PLUS:op ident:rightval
 {: RESULT=new binaryOpNode(leftval,
      sym.PLUS, rightval,
      op.linenum,op.colnum); :}

90 CS 536 Fall 2002^©

| exp:leftval MINUS:op ident:rightval
 {: RESULT=new binaryOpNode(leftval,
           sym.MINUS,rightval,
           op.linenum,op.colnum); :}
| ident:i
 {: RESULT = i; :}
;
ident::= IDENTIFIER:i
 {: RESULT = new nameNode(
   new identNode(i.identifierText,
                i.linenum,i.colnum),
   exprNode.NULL,
   i.linenum,i.colnum); :}
;

91 CS 536 Fall 2002^©

Let’s parse { a = b ; } First, a is parsed using

ident::= IDENTIFIER:i
 {: RESULT = new nameNode(
   new identNode(i.identifierText,
                i.linenum,i.colnum),
   exprNode.NULL,
   i.linenum,i.colnum); :}

We build

nameNode

identNode nullExprNode a

92 CS 536 Fall 2002^©

Next, a is parsed using

ident::= IDENTIFIER:i
 {: RESULT = new nameNode(
   new identNode(i.identifierText,
                i.linenum,i.colnum),
   exprNode.NULL,
   i.linenum,i.colnum); :}

We build

nameNode

identNode nullExprNode b

93 CS 536 Fall 2002^©

Then b’s subtree is recognized as an exp:

| ident:i
 {: RESULT = i; :}

Now the assignment statement is recognized:

stmt::= ident:id ASG exp:e SEMI
 {: RESULT=
       new asgNode(id,e,
           id.linenum,id.colnum);
 :}

We build

nameNode

identNode nullExprNode a nameNode

identNode nullExprNode b asgNode

94 CS 536 Fall 2002^©

The stmts→ λ production is matched(indicatingthat thereareno more statements in the program). CUP matches

stmts::=
 {: RESULT= stmtsNode.NULL; :}

and we build

Next, stmts→ stmt stmts is matched using

stmts::= stmt:s1  stmts:s2
 {: RESULT=
     new stmtsNode(s1,s2,
       s1.linenum,s1.colnum);
 :}

nullStmtsNode

95 CS 536 Fall 2002^©

This builds

As the last step of the parse, the parser matches program→ { stmts } using the CUP rule

prog::= LBRACE:l stmts:s RBRACE
 {: RESULT=
     new csxLiteNode(s,
          l.linenum,l.colnum); :}
;

nameNode

identNode nullExprNode a nameNode

identNode nullExprNode b asgNode stmtsNode

nullStmtsNode

96 CS 536 Fall 2002^©

ThefinalASTreturnedbythe parseris

nameNode

identNode nullExprNode a nameNode

identNode nullExprNode b asgNode stmtsNode

nullStmtsNode csxLiteNode